

## Gramática da linguagem PL/0

Fonte: Wirth, Niklaus. (1976). *Algorithms + Data Structures = Programs*. Prentice Hall, Inc.

(mais detalhes em <https://en.wikipedia.org/wiki/PL/0>)

```
<programa> ::= <bloco> .
<bloco> ::= <declaracao> <comando>
<declaracao> ::= <constante> <variavel> <procedimento>
<constante> ::= CONST ident = numero <mais_const> ; | λ
<mais_const> ::= , ident = numero <mais_const> | λ
<variavel> ::= VAR ident <mais_var> ; | λ
<mais_var> ::= , ident <mais_var> | λ
<procedimento> ::= PROCEDURE ident ; <bloco> ; <procedimento> | λ
<comando> ::= ident := <expressao>
                | CALL ident
                | BEGIN <comando> <mais_cmd> END
                | IF <condicao> THEN <comando>
                | WHILE <condicao> DO <comando>
                | λ
<mais_cmd> ::= ; <comando> <mais_cmd> | λ
<expressao> ::= <operador_unario> <termo> <mais_termos>
<operador_unario> ::= - | + | λ
<termo> ::= <fator> <mais_fatores>
<mais_termos> ::= - <termo> <mais_termos> | + <termo> <mais_termos> | λ
<fator> ::= ident | numero | ( <expressao> )
<mais_fatores> ::= * <fator> <mais_fatores> | / <fator> <mais_fatores> | λ
<condicao> ::= ODD <expressao>
                | <expressao> <relacional> <expressao>
<relacional> ::= = | <> | < | <= | > | >=
```

Além disso:

- comentários são de única linha, entre chaves { }
- identificadores são formados por letras e dígitos, começando por letra
- só há números inteiros, formados por um ou mais dígitos (entre 0 e 9)

Exemplos de programas:

### Fatorial (sem procedimento)

```
VAR n, fat;
BEGIN
    n:=4;
    fat:=1;
    WHILE n>1 DO
        BEGIN
            fat:=fat*n;
            n:=n-1;
        END
    END.
```

### Fatorial (com procedimento)

```
VAR n, fat;
PROCEDURE fatorial;
BEGIN
    fat:=1;
    WHILE n>1 DO
        BEGIN
            fat:=fat*n;
            n:=n-1;
        END
    END;
BEGIN
    n:=4;
    CALL fatorial
END.
```